

Wikidataを例とした SPARQLの例題＋演習課題

SPARQLによるRDFの検索

■ SPARQL

- RDFデータに対するクエリ言語
- 「指定したグラフ構造」に一致するトリプルを検索する

重要

■ 最も基本的な検索

```
select ?s ?p ?o
```

←返す要素

```
where {
```

```
  ?s ?p ?o .
```

?x (x: 任意の文字列) は変数を表す

←検索するグラフのパターン

```
}
```

```
LIMIT 100
```

「.」(ピリオド)
を忘れない

↑取得する数の制限

このパターンを変
えることで、欲しい
データを取得する

SPARQLによる DBpedia Japaneseの検索例



■ 「各都道府県で生まれた政治家の数」を調べる

PREFIX rdfs: <<http://www.w3.org/2000/01/rdf-schema#>>

PREFIX dbpedia-owl: <<http://dbpedia.org/ontology/>>

PREFIX rdf: <<http://www.w3.org/1999/02/22-rdf-syntax-ns#>>

PREFIX dbpedia-ja: <<http://ja.dbpedia.org/resource/>>

PREFIX category-ja: <<http://ja.dbpedia.org/resource/Category:>>

select distinct ?pref (count(?s) AS ?c) where {

 ?pref rdf:type dbpedia-owl:Place.

 ?pref dbpedia-owl:wikiPageWikiLink category-ja:日本の都道府県.

 ?s rdf:type dbpedia-owl:Politician;

 dbpedia-owl:birthPlace ?pref.

}GROUP BY ?pref

ORDER BY ?c

SPARQLエンドポイント

(<http://ja.dbpedia.org/sparql>)

実行結果

(解説) **Qiita**: DBpediaを使った都道府県別ランキング

<http://qiita.com/koujikozaiki/items/439fa7ce3e28b738fe10>

SPARQLによる DBpedia Japaneseの検索例



■ 「各都道府県で生まれた芸人の数」を調べる

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX dbpedia-ja: <http://ja.dbpedia.org/resource/>

PREFIX category-ja: <http://ja.dbpedia.org/resource/Category:>

```
select distinct ?pref (count(?s) AS ?c) where {  
  ?pref rdf:type dbpedia-owl:Place.  
  ?pref dbpedia-owl:wikiPageWikiLink category-ja:日本の都道府県.  
  ?s rdf:type dbpedia-owl:Comedian;  
    dbpedia-owl:birthPlace ?pref.  
}GROUP BY ?pref  
ORDER BY ?c
```

SPARQLエンドポイント

(<http://ja.dbpedia.org/sparql>)

実行結果

ここを,

Politician → **Comedian**
に変わるだけ！

SPARQLによる検索例

Wikidataを用いたランキング



■ 「日本の政治家の出身大学」ランキング

→ 実行結果

```
select ?univ ?univl (count(?s) As ?c) where{  
  ?univ wdt:P31/wdt:P279* wd:Q3918. #大学の一覧取得  
  ?s wdt:P27 wd:Q17; #国籍＝日本  
    wdt:P106 wd:Q82955; #職業＝政治家  
    wdt:P69 ?univ. #出身大学  
OPTIONAL{  
  ?s rdfs:label ?name. FILTER(lang(?name)="ja")  
  ?univ rdfs:label ?univl . FILTER (lang(?univl) = "ja") .  
}  
}GROUP BY ?univ ?univl  
ORDER BY DESC(?c)  
LIMIT 100
```

政治家の出身大学ランキング
<https://qiita.com/koujikozaiki/items/a049e2ac1051e0e43be6>

SPARQLによる検索例

Wikidataを用いたランキング



■ 「日本の総理大臣の出身大学」ランキング

→ **実行結果**

```
select ?univ ?univl (count(?s) As ?c) where{  
?univ wdt:P31/wdt:P279* wd:Q3918. #大学の一覧取得  
?s wdt:P27 wd:Q17; #国籍＝日本  
   wdt:P106 wd:Q82955; #職業＝政治家  
   wdt:P39 wd:Q274948; #公職＝内閣総理大臣  
   wdt:P69 ?univ. #出身大学
```

この1行を
追加する

```
OPTIONAL{  
  ?s rdfs:label ?name. FILTER(lang(?name)="ja")  
  ?univ rdfs:label ?univl . FILTER (lang(?univl) = "ja") .  
}  
}GROUP BY ?univ ?univl  
ORDER BY DESC(?c)  
LIMIT 100
```

政治家の出身大学ランキング
<https://qiita.com/koujikozaiki/items/a049e2ac1051e0e43be6>

SPARQLクエリ演習

- WikidataのSPARQLエンドポイント(検索用API)
<https://query.wikidata.org/>
を使った, SPARQLクエリの演習.
- クエリ例は, すべて,
<https://github.com/koujikozaiki/SPARQLexamples/blob/master/WikidataEx1.md>
から試せる.

■ 例題の解説

- スライドを使って例題のクエリを解説する
- 解説後，例題のクエリを実行してみる


■ 演習課題

- 例題のクエリの「**一部を変更**」し，演習課題で指定したクエリを作成する
- 作成したクエリは，「**SPARQL演習20190708.txt**」に，**記録**する．各クエリごとに，下記の3点を記録しておくとい
 - クエリ内容の**説明**
 - 作成した「**クエリ文**」
 - 実行用の「**短縮URL**」


[解説動画はこちら](#)

検索例1: 主語と述語を指定して「目的語」を取得 “<主語>の<述語>は何?”

検索するグラフパターン



SPARQLでの記述



検索例1: 主語と述語を指定

- 例1)「大阪電気通信大学」(主語)の「位置する行政区」(述語)となる**目的語(?o)**を取得する

```
select ?o  
where {  
  wd:Q7105556 wdt:P131 ?o .  
}
```

大阪電気
通信大学
(主語)

位置する
行政区
(述語)

目的語
(変数)

検索例1: 主語と述語を指定

- 例1-1)「大阪電気通信大学」(主語)の「設立」(述語)となる**目的語(?o)**を取得する

```
select ?o  
where {  
  wd:Q7105556 wdt:P571 ?o .  
}
```

大阪電気
通信大学
(主語)

設立
(述語)

目的語
(変数)

※ 述語を変えるといろんな
目的語が取得できる

検索例1: 主語と述語を指定



- 例1-2)「大阪電気通信大学」(主語)の「位置する行政区」(述語)となる**目的語(?o)**を取得する

※ 検索結果がデータのIDとなる場合, **下記の記述を追加することで「ラベル」をあわせて取得可能**

Wikidata
のみの
仕様

```
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
```

```
select ?o ?oLabel
```

元の変数名 + Label

```
where {
```

```
  wd:Q651233 wdt:P159 ?o .
```

```
  SERVICE wikibase:label { bd:serviceParam  
    wikibase:language "[AUTO_LANGUAGE],ja". }  
}
```

英語にしたいならen

例1-3: 複数の述語を指定して, 目的語を取得する

- ※複数行ならべると, まとめて目的語を取得できる.
(変数名は変える)
- 「大阪電気通信大学」(主語)の「位置する行政区」(述語)となる目的語(?o), および「設立」(述語)となる目的語(?o2)を取得

```
select ?o ?oLabel ?o2
```

```
where {
```

```
wd:Q7105556 wdt:P131 ?o .
```

位置する行政区

```
wd:Q7105556 wdt:P571 ?o2 .
```

設立

```
SERVICE wikibase:label { bd:serviceParam  
wikibase:language "[AUTO_LANGUAGE],ja". }  
}
```

【演習1】主語と述語を指定して「目的語」を取得する



■ 演習1-a:「述語」を変えてみる

- 「検索例1-1, 2, 3」の「**述語**」を変えて, 「**大阪電気通信大学**」(**主語**)のいろんな**情報(目的語?o)**を取得してみる
- **述語のID**は「大阪電気通信大学」の**Wikidataのページ**を見て探す.
→プロパティにマウスを持っていくと表示される「**P〇〇**」の番号を使えばよい.

■ 演習1-b:「主語」を変えてみる

- 「検索例1-1, 2, 3」の「**主語**」を変えて, いろんな主語の**情報(目的語?o)**を取得してみる
- **主語のID**は, 探したいデータの「Wikidataのページ」(Wikipediaのページの右側にあるメニューから「**ウィキデータ項目**」の**リンク**で表示できる)を見て探す.
→ページ上部の「**Q〇〇〇〇〇**」の番号を使えばよい.

[解説動画はこちら](#)

検索例2:
述語と目的語を指定して「主語」を取得
“<述語>が<目的語>となる<主語>は？”

検索するグラフパターン



SPARQLでの記述

?s <位置する行政区> <寝屋川市>

検索例2: 述語と目的語を指定

- 検索例2-1「位置する行政区」(述語)が「寝屋川市」(目的語)となる「主語(?s)」の一覧を取得する
 - 「主語」の取得は検索結果が多数となる場合が多い

```
select ?s ?sLabel
```

```
where {
```

位置する行政区(述語)

```
?s wdt:P131 wd:Q389633 .
```

寝屋川市(目的語)

```
SERVICE wikibase:label { bd:serviceParam  
wikibase:language "[AUTO_LANGUAGE],ja". }  
}
```

```
LIMIT 100
```

結果が多いので、件数の制限をかける

検索例2: 述語と目的語を指定

- 例) 2-2: 「分類が“大学”」となる主語(?s)を取得
 - 分類(instance-of)を使うと同じ種類のデータ一覧が取得できる

Wikidataのみ

他のLODでは,
instance-ofではなく, **rdf:type**
が用いられることが多い.

```
select ?s ?sLabel
```

```
where {
```

分類(instance-of)

大学(目的語)

```
?s wdt:P31 wd:Q3918 .
```

```
SERVICE wikibase:label { bd:serviceParam  
wikibase:language "[AUTO_LANGUAGE],ja". }  
}
```

```
LIMIT 100
```

【演習2】述語と目的語を指定して 「主語」の一覧を取得する



- 演習2-a:「目的語」を変えて, いろんな種類(クラス)のデータ一覧を取得してみる
 - 「検索例2-2」の「目的語」となるクラスを変える
クラスのIDは適当なデータの「Wikidataのページ」で
instance-of (P31)の目的語 を調べると良い.
- 演習2-b:「述語」と「目的語」の組み合わせを変えて, いろんなデータ一覧を取得してみる
 - 「検索例2-1」の「述語」や「目的語」を変える


検索例3:

解説動画はこちら

「主語」の一覧の「絞り込み」

“「... という条件を満たす」となる<主語>は？”

検索するグラフパターン



SPARQLでの記述

?s <instance-of> <大学>.

?s <国> <日本>.

検索例3 :「主語」の一覧の「絞り込み」

- 検索例3-1「大学の一覧(主語)」を「国(述語)」の「目的語(?country)」と共に取得する
 - 同じ主語のトリプルを続けることで、合わせて取得したい内容を指定する. ※2行以上の指定も同様に可

```
select ?s ?sLabel ?country ?countryLabel
where {
  ?s wdt:P31 wd:Q3918 .
  ?s wdt:P17 ?country .
  SERVICE wikibase:label { bd:serviceParam
    wikibase:language "[A...LANGUAGE],ja". }
}
LIMIT 100
```

例2-2)「分類」が「大学」
の主語一覧

取得した主語(?s)の「国(述語)」
の「目的語(?country)」

検索例3 :「主語」の一覧の「絞り込み」

- 検索例3-2「大学の一覧」を「国(述語)が日本(目的語)」のものに絞り込む
 - 同じ主語のトリプルを続けることで、主語が満たす条件を指定する。 ※2行以上の指定も同様に可

```
select ?s ?sLabel
```

```
where {
```

```
  ?s wdt:P31 wd:Q3918 .
```

```
  ?s wdt:P17 wd:Q17 .
```

例2-2)「分類」が「大学」
の主語一覧

```
SERVICE wikibase:label { bd:serviceParam
```

```
wikibase:language "ja"
```

```
}
```

```
LIMIT 100
```

取得した主語(?s)の「国(述語)」の「目的語(?country)」が「**日本(wd:Q17)**」

検索例3 :「主語」の一覧の「絞り込み」



- 検索例3-2「大学の一覧」を「国(述語)が日本(目的語)」のものに絞り込む
 - 「国(述語)」の「目的語」も合わせて取得し、「国＝日本」で絞り込まれていることを確認する。

```
select ?s ?sLabel ?country ?countryLabel
where {
  ?s wdt:P31 wd:Q3918 .
  ?s wdt:P17 ?country .
  ?s wdt:P17 wd:Q17 .
  SERVICE wikibase:label { bd:serviceParam
    wikibase:language "[AUTO_LANGUAGE],ja". }
}
LIMIT 100
```

検索例3 :「主語」の一覧の「絞り込み」

- 検索例3-3 「日本にある大学の一覧」を「設立日」と共に取得する
 - 「絞り込んだ主語(?s)」の一覧と、同じ主語のトリプルを続けることで、その主語が持つ情報を取得できる。

```
select ?s ?sLabel ?o  
where {
```

検索例3-2 「日本にある
大学の一覧」の取得

```
?s wdt:P31 wd:Q3918 . # ?sの「分類」が「大学」
```

```
?s wdt:P17 wd:Q17 . # ?sの「国」が「日本」
```

```
?s wdt:P571 ?o . # ?sの「設立」を?oとする
```

```
SERVICE wikibase:label { bd:serviceParam  
wikibase:language "[AUTO_LANGUAGE],ja". }  
}LIMIT 100
```

※クエリ内の#以降は、「コメント」

検索例3 :「主語」の一覧の「絞り込み」

- 検索例3-3-a「日本にある大学の一覧」を「設立日」と共に取得し、「**設立日**」で並び替え
 - **ORDEY BY**: 並び替えの関数

```
select ?s ?sLabel ?o
where {
  ?s wdt:P31 wd:Q3918 . # ?Sの「分類」が「大学」
  ?s wdt:P17 wd:Q17 .   # ?sの「国」が「日本」
  ?s wdt:P571 ?o .      # ?sの「設立」を?oとする
  SERVICE wikibase:label { bd:serviceParam
    wikibase:language "[AUTO_LANGUAGE],ja". }
} ORDER BY ?o
LIMIT 100
```

?o(設立日)で並び替え
※**DESC(?o)**とする「降順」に

【演習3】

いろんなデータの一覧を取得してみる



■ 演習3-a

:「目的語」を変えて, さまざまな「絞り込み」を試す

- 「検索例3-2」の「述語」と「目的語」の組み合わせを変え, 様々な条件で絞り込んだ「大学の一覧」を取得する

■ 演習3-b

:いろんなデータ一覧を取得する

- 「検索例3-1,2,3」を変更して, いろんなデータ一覧を取得してみる
- 演習2-aや演習2-bに, 「主語の条件」を追加して, データ一覧を絞り込む